Infrarot-Transceiver für den PC (Teil 2, Elektor Dezember 2001)

Die folgenden Teile aus dem Artikel konnten aus Platzgründen nicht in Elektor gedruckt werden und stehen deshalb als Download-Datei bereit. Es handelt sich hier um Grundlagen zur seriellen Datenübertragung, die unabhängig von der IR-Übertragung auch für drahtgebundene Strecken gilt. Die Beispiele wurden in großen Teilen aus dem folgenden Buch übernommen:

B. Kainka, Elektronik am PC, Visual Basic in der Praxis, Elektor-Verlag 2001

Serielle Übertragung

Zur Nullmodemverbindung

Das erste Programmbeispiel (Listing 4) zur Datenübertragung überträgt bei jedem Kommando nur ein Byte. Und zwar wird in diesem Fall die Position eines Schiebereglers als Zahlenwert zwischen 0 und 255 an den Partner übermittelt. Umgekehrt wird jedes empfangene Byte in einem Textfenster angezeigt.
Private Sub Form_Load()

 OPENCOM "COM2:1200,N,8,1"

End Sub

Private Sub Form_Unload(Cancel As Integer)

  CLOSECOM

End Sub

Private Sub Command1_Click()

  d = HScroll1.Value

  SENDBYTE d

End Sub

Private Sub Timer1_Timer()

  d = READBYTE

  If d > -1 Then Text1.Text = Str$(d)

End Sub

Listing 4

Senden und Empfangen einzelner Bytes (RSbyte1.vbp)

Universelles Terminalprogramm
Für die Arbeit mit der RS232-Schnittstelle braucht man oft ein Terminal-Programm, das empfangene Daten anzeigt und die Ausgabe von Daten erlaubt. Unter Windows hat man bereits das Hyperterminal, mit dem sich z.B. Daten über ein Modem übertragen lassen. Hier soll jedoch ein TerminalprogrammXE "Terminalprogramm" vorgestellt werden, das speziell für allgemeine Entwicklungsaufgaben möglichst universell eingesetzt werden kann.

Der Anwender kann den Open-String direkt eingeben und die Schnittstelle mit den gewünschten Daten öffnen. In solchen Fällen, wo die Übertragungsparameter eines Geräts nicht genau bekannt sind, lassen sie sich schnell ausprobieren. Nach jeder Änderung des Open-Strings und einer Betätigung der Schaltfläche "New" wird die Schnittstelle geschlossen und mit den neuen Parametern erneut geöffnet.

Die selben seriellen Daten lassen sich auf zwei völlig unterschiedliche Arten interpretieren. Einmal handelt es sich um Steuerdaten im Byte-Format, also einfach um Zahlenwerte im Bereich 0 bis 255. Man spricht hier auch vom Binärformat. Ein Beispiel dafür war die Übertragung der Position eines Schiebereglers aus dem vorigen Abschnitt. Zahlreiche Messgeräte, Interfaces und Geräte wie Computer-Mäuse oder andere Zeigegeräte verwenden diesen Modus.

[image: image1.png]eminal [-[ofx]

OpenCom  [COM2:1200.N.8.1 New

SendByte ReadByte

=] [e2 8 50 51 50 13 84 101 |
14 103 105 110 57 108 13

L o L o

SendTent Read Tent
ez = [z |
Termina Temina

o L
Clear

DR [ ATS [Cis DSR [DCD [ Al





Bild 1 zum Terminalprogramm: Textübertragung mit dem Terminalprogramm

Die andere Art der Interpretation übertragener Daten ist die als Text. Textzeichen sind immer auch Bytes. Das Bild 1 zum Terminalprogramm zeigt die Übertragung von Textzeichen. Die Leitungen TXD und RXD wurden für diesen Test direkt verbunden, sodass der eingegebene Text an der selben Schnittstelle wieder empfangen wurde. Der ASCII-Zeichensatz definiert bekanntlich, welches Textzeichen durch welchen Zahlenwert repräsentiert wird. Der Zahlenwert 65 steht für das große A, 49 steht für die Ziffer 0. Wer Texte überträgt, kümmert sich normalerweise nicht um den ASCII-CodeXE "ASCII-Code". Es gibt jedoch Fälle, in denen eine genauere Betrachtung sinnvoll ist. Wer z.B. vor der Aufgabe steht ein Messgerät mit serieller Schnittstelle über ein Programm abzufragen, findet sich oft in einem Dschungel unklarer Auskünfte in der Betriebsanleitung wieder. Oft werden Textdaten und binäre Daten gemischt eingesetzt.

Das Terminalprogramm nach Listing 5 unterstützt von vornherein beide Interpretationen von Zeichen. Eingehende Daten werden gleichzeitig als Bytes und als Textzeichen dargestellt. Man sieht dann sehr schnell, was gemeint ist. Auch beim Senden hat man die Möglichkeit, beide Formate zu verwenden. Im unteren Fenster tippt man einfach Texte ein, die Zeichen für Zeichen übertragen werden. Im oberen Fenster dagegen werden Zahlenwerte zwischen 0 und 255 eingegeben und mit der Return-Taste bestätigt.

Dim n

Private Sub Check1_Click()

  If Check1.Value Then DTR 1 Else DTR 0

End Sub

Private Sub Check2_Click()

  If Check2.Value Then RTS 1 Else RTS 0

End Sub

Private Sub Command1_Click()

  CLOSECOM

  OPENCOM Text1.Text

  DTR 0

  Check1.Value = False

  RTS 0

  Check2.Value = False

End Sub

Private Sub Command2_Click()

 Text2.Text = ""

 Text3.Text = ""

 Text4.Text = ""

 Text5.Text = ""

End Sub

Private Sub Form_Load()

  OPENCOM "COM2:1200,N,8,1"

  DTR 0

  Check1.Value = False

  RTS 0

  Check2.Value = False

End Sub

Private Sub Form_Unload(Cancel As Integer)

  CLOSECOM

End Sub

Private Sub Text2_Change()

  If Text2.Text > "" Then Char = Asc(Right$(Text2.Text, 1))

  If Char = 10 Then

    Byt$ = (Right$(Text2.Text, 5))

    dat = Val(Byt$)

    dat = dat And 255

    SENDBYTE dat

  End If

End Sub

Private Sub Text4_KeyPress(KeyAscii As Integer)

  SENDBYTE KeyAscii

End Sub

Private Sub Timer1_Timer()

  Do

    dat = READBYTE

    If dat > -1 Then

      Text3.Text = Text3.Text + Str$(dat) + " "

       n = n + 1

       If n = 8 Then n = 0: Text3.Text = Text3.Text + Chr(13) + Chr(10)

       Text3.Refresh

    End If

    If (dat > -1) And (dat <> 13) And (dat <> 19) Then

      Text5.Text = Text5.Text + Chr$(dat)

    End If

    If dat = 13 Then

      Text5.Text = Text5.Text + Chr$(13) + Chr$(10)

      Text5.Refresh

    End If

  Loop Until dat = -1

  If CTS = 1 Then Check3.Value = 1 Else Check3.Value = 0

  If DSR = 1 Then Check4.Value = 1 Else Check4.Value = 0

  If DCD = 1 Then Check5.Value = 1 Else Check5.Value = 0

  If RI = 1 Then Check6.Value = 1 Else Check6.Value = 0

End Sub

Listing 5 

Das RS232-Terminal (Terminal.vbp)

Beim Senden von Text im unteren Eingabefenster (Text4) werden Textzeichen unverändert übertragen, die den einzelnen Tasteneingaben entsprechen. Visual Basic liefert dazu den Zeichenwert in der Variablen KeyAscii, der direkt an SENDBYTE weitergegeben werden kann. Auch die Returntaste liefert ein Zeichen, nämlich das Line-Feed-Zeichen 10.

Die Interpretation von Zahlenwerten im oberen Textfenster (Text2) ist etwas aufwendiger. Das Programm muss zunächst abwarten, bis der Anwender eine Zahleneingabe mit der Return-Taste bestätigt. Der Text enthält dann an letzer Stelle das LF-Zeichen 10. Nun werden die letzen vier Zeichen abgetrennt und von einem Text in einen Zahlenwert umgewandelt. Dieser sollte nun ein Wert zwischen 0 und 255 sein. Zur Sicherheit gegen Fehleingaben von größeren Zahlen wird der übergebene Wert mit "AND 255" begrenzt.

Der Empfang von Zeichen erfolgt in einer Timer-Prozedur. Alle 100 ms wird nachgesehen, ob Zeichen eingetroffen sind. Damit es zu keinem Stau kommt, werden jedesmal alle angekommenen Zeichen ausgelesen und verarbeitet, bis READBYTE das Ergebnis -1 liefert, also auf einen leeren Puffer trifft. Die empfangenen Zeichen werden auf den beiden Ausgabefenstern Text3 und Text5 dargestellt.

Bei der Ausgabe im Zahlenformat im Textfernster Text3 werden die Zahlenwerte in Zeilen zu jeweils acht Bytes angeordnet. Danach erfolgt jeweils ein Zeilenumbruch mit den Sonderzeichen Line Feed (LF, ASCII 10) und Carriage Return (CR, ASCII 13).

Die Textausgabe im Fenster Text5 muss ebenfalls Zeilenumbrüche besonders behandeln. Das Sonderzeichen LF wird abgefangen und durch LF und CR ersetzt. Alle anderen Zeichen werden direkt an den Text angehängt. Im Ergebnis erscheint ein Text in der Ausgabe genauso, wie er im Eingabefenster eingegeben wurde.

Das Programm lässt sich als universelles Werkzeug zur Erprobung serieller Verbindungen einsetzen. Eine weitere Hilfestellung bietet die direkte Zugriffsmöglichkeit auf alle Handshakeleitungen im selben Programm. Manche Geräte benötigen bestimmte Zustände an DTR oder RTS. So erhält z.B. die serielle PC-Maus ihre Betriebsspannung über die RTS-Leitung. Man kann den Datentransfer einer Maus mit dem Terminalprogramm untersuchen. Bild 2 zum Terminalprogramm zeigt ein Ergebnis.

Bild 2 zum Terminalprogramm: Empfang binärer Daten von einer seriellen Maus

